


Multi-frequency C-V Measurements and Doping Profile Analysis of Semiconductors

— HP 4194A Application Information —



**HP 4194A Impedance/Gain-Phase Analyzer
with HP 41941A Impedance Probe Kit**

Introduction

The HP 4194A Impedance/Gain-Phase Analyzer is the best choice for multi-frequency C-V measurements of semiconductors, thanks to its wide measurement frequency range (100kHz to 100MHz when used with the HP 41941A/B Impedance Probe Kit), precise measurement capability—even if the 1.5m/3m long probe is used, its optimized measurement performance and its ability to display directly plot measurement results. This application information describes how the HP 4194A contributes to new material and process development semiconductor labs.

Problems and Solutions Offered by the HP 4194A

Multi-frequency

As the operating frequency of devices has increased above 1MHz, the test frequency range for C-V measurements has increased accordingly.

Low frequency (such as 100kHz) measurements are required when measuring large diameter wafers. Because the thinness of large diameter wafers result in a higher series resistance at higher frequencies. (Refer to Figure 1), especially for high resolution oxide-layer capacitance (C_{ox}) and depletion-layer capacitance (C_d) measurements.

—) The HP 4194A's measurement frequency, 100Hz to 40MHz (standard), or 10kHz to 100MHz when using the HP 41941A/B Impedance Probe Kit, covers both of these measurement requirements.

Extension Cable to Prober Station

An extension cable from an LCR meter to a prober station (approximately 1 to 3m) causes measurement errors, and accurate measurements at frequencies above 1MHz may become impossible. By using the HP 41941A/B Impedance Probe Kit with the HP 4194A, you can make precise measurements up to 100MHz using a 1.5m or 3m long impedance probe, without the measurement errors caused by an extension cable.

C-V Curve Plotting and Calculation for Doping Profile

In the past, an external computer was needed to plot a C-V curve and to calculate a doping profile ($N(w)$ vs. w^*). Doping profiles from C-V data are most commonly used for the evaluation of semiconductor device structures and processing methods.

—) The HP 4194A's built-in color CRT can simultaneously display a C-V curve and doping profile. The Auto Sequence Program (ASP) function can be used to display these parameters quickly and automatically. Figure 3 shows a sample program. A quick hardcopy is also available (see Figures 5 and 6).

* $N(w)$: Doping Profile

w: Depth




Figure 1 A Model of MOS Structure




Figure 2 Measurement Setup
(HP 4194A + HP 41941A/B)

```

10 RST
20 CMT "**** MULTI FREQUENCY C-V CHARACTERISTIC ****"
30 ! FNC1;IMP12      ! IMPEDANCE
40  FNC3;IMP12      ! IMP. WITH Z PROBE
50 UNIT0;SWM2;OPN1;SHT1
60 BEEP
70 Z=0
80 DISP "COMPENSATION? / Y->INPUT <1>"
90 PAUSE
100 IF Z=1 THEN GOSUB 730
110 R10= 8.854E-14    ! FREE SPACE PERMITIVITY
120 R12= 0.001      ! AREA OF THE GATE
130 R11=-1.602E-19   ! MAGNITUDE OF ELECTRONIC CHARGE
140 R13= 11.7        ! DIELECTRIC CONSTANT OF SI
150 SWP2;START=-5 V;STOP=5 V;STEP=.2 V
160 R51=NOP          ! NUMBER OF POINT
170 R50=R51-1        !
180 R52=20000        ! HOLD TIME (MSEC)
190 R53=100          ! DELAY TIME (MSEC)
200 DISP "CONNECT TEST DEVICE"
210 PAUSE
220 ! *** MEAS. START ***
230 FOR R1=1 TO 3
240   IF R1=1 THEN FREQ=100 KHZ;GOTO 270 ! SET SPOT FREQUENCY
250   IF R1=2 THEN FREQ= 1 MHZ;GOTO 270
260   FREQ= 10 MHZ
270   DTIME=R52;TRGM2
280   SWTR6
290   TRIG
300   DTIME=R53
310   TRGM1;MCF1;MKMXA
320   IF R1=1 THEN E=A;R5=MKRA;GOTO 350 ! 100 KHZ DATA TO E
330   IF R1=2 THEN F=A;R6=MKRA;GOTO 350 ! 1 MHZ DATA TO F
340   G=A;R7=MKRA ! 10 MHZ DATA TO G
350 NEXT R1
360 AMAX=1;BMAX=1;AMIN=0;BMIN=0;A=G/R7;B=F/R6;SPSTR;A=E/R5;SPA1;CPYM3
370 DISP "A=100 KHZ/B=1 MHZ/C=10 MHZ"
380 BEEP
390 PAUSE
400 CMT "          **** DOPING PROFILE ****"
410 R20=2/(R11*R10*R13*(R12**2))
420 R21=R12*R10*R13
430 FOR R1=1 TO R50
440   R2=R1+1
450   H(R1)=(1/(E(R2)*E(R2))-1/(E(R1)*E(R1)))/STEP
460   H(R1)=R20/H(R1) ! 100 KHZ N(W)
470   I(R1)=(1/(F(R2)*F(R2))-1/(F(R1)*F(R1)))/STEP
480   I(R1)=R20/I(R1) ! 1 MHZ N(W)
490   J(R1)=(1/(G(R2)*G(R2))-1/(G(R1)*G(R1)))/STEP
500   J(R1)=R20/J(R1) ! 10 MHZ N(W)
510 NEXT R1
520 H(R51)=H(R50);I(R51)=I(R50);J(R51)=J(R50);H=ABS(H);I=ABS(I);J=ABS(J)
530 RA=R21*(1/E-1/R5) ! 100 KHZ W
540 RB=R21*(1/F-1/R6) ! 1 MHZ W
550 RC=R21*(1/G-1/R7) ! 10 MHZ W
560 DSP2;BSC2;DPAB0
570 GOSUB 820

```

```

580 DPAB1
590 FOR R1=1 TO 3
600 IF R1=1 THEN A=RA;B=H ;GOTO 630
610 IF R1=2 THEN A=RB;B=I ;GOTO 630
620         A=RC;B=J
630 AMAX=R91;AMIN=R92;BMIN=R93;BMIN=R94
640 BEEP
650 IF R1=1 THEN CMT "      **** DOPING PROFILE AT 100KHZ ****" ;GOTO 680
660 IF R1=2 THEN CMT "      **** DOPING PROFILE AT 1MHZ   ****" ;GOTO 680
670         CMT "      **** DOPING PROFILE AT 10MHZ  ****"
680 DISP " ---> PRESS (CONT) "
690 PAUSE
700 NEXT R1
710 DISP "COMPLETED !"
720 END
730 ! **** COMPENSATION SUBROUTINE ****
740 BEEP
750 DISP "OPEN !"
760 PAUSE
770 ZOPEN
780 DISP "SHORT !"
790 PAUSE
800 ZSHRT
810 RETURN
820 ! **** AUTO SCALE SUBROUTINE ****
830 FOR R90=1 TO 3
840     IF R90=1 THEN A=E ;B=H ;GOTO 870
850     IF R90=2 THEN A=F ;B=I ;GOTO 870
860         A=G ;B=J
870     AUTO
880     IF R90=1 THEN R91=AMAX ;R92=AMIN ;R93=BMAX ;R94=BMIN ;GOTO 910
890     IF R90=2 THEN R81=AMAX ;R82=AMIN ;R83=BMAX ;R84=BMIN ;GOTO 910
900         R71=AMAX ;R72=AMIN ;R73=BMAX ;R74=BMIN
910 NEXT R90
920     IF R91<R81 THEN R91=R81
930     IF R91<R71 THEN R91=R71
940     IF R92>R82 THEN R92=R82
950     IF R92>R72 THEN R92=R72
960     IF R93<R83 THEN R93=R83
970     IF R93<R73 THEN R93=R73
980     IF R94>R84 THEN R94=R84
990     IF R94>R74 THEN R94=R74
1000 RETURN

```

Figure 3 ASP Program Listing




Figure 4 ASP Flow Chart




Figure 5 Example C-V Curve Display




Figure 6 Example Doping Profile Display

ASP UPLOAD/DOWNLOAD PROGRAM

The HP 4194A stores ASP programs in its 20K byte internal nonvolatile memory. Data can be stored for 2000 hours after the battery used in the nonvolatile memory circuit has been charged for 48 hours.

However, ASP programs should be backed up in the following cases:

- Many ASP programs for device testing are used.
- One HP 4194A is used by many people who have their own ASP programs.

```
10 1 4194A ASP UPLOAD PROGRAM
20 DIM Prog$(300)(100)
30 DIM S$(20),Name$(10)
40 Add=717
50 !
60 BEEP
70 INPUT "Input ASP File number",No
80 OUTPUT Add;"LOAD";No
90 OUTPUT Add;"EDIT"
100 OUTPUT Add;"QUIT"
110 !
120 BEEP
130 INPUT "Enter File name to be stored",Name$
140 OUTPUT Add;"CPYM2"
150 OUTPUT Add;"COPY"
160 ENABLE INTR 7:2048
170 ON INTR 7 GOTO Store_disk
180 X=1
190 ENTER Add USING "X,20A":S$
200 ENTER Add:Prog$(X)
210 X=X+1
220 GOTO 200
230 Store_disk:1
240 DISABLE INTR 7
250 FOR I=1 TO X-1
260 Prog$(I)="PROG"&CHR$(39)&Prog$(I)&CHR$(39)
270 PRINT Prog$(I)
280 NEXT I
290 L=X-1
300 !
310 CREATE BDAT Name$,L+1,100
320 ASSIGN @File TO Name$
330 FOR I=1 TO L
340 OUTPUT @File,I:Prog$(I)
350 NEXT I
360 OUTPUT @File,L+1;"Eof"
370 BEEP
380 DISP " Up Load Complete"
390 END
```

Program 7 BASIC Program for ASP Upload

APPENDIX

Reference

Application Note 339-11

Filter Test for Production and Incoming Inspection using the HP 4194A

Programs 7 and 8 are useful utility programs for the above situations.

Program 7 is a BASIC program to upload and store ASP programs to the computers mass storage device, (floppy disc). Uploading is performed using the HP 4194A's COPY command, which prints out a listing of an ASP program and measurement data on a printer.

Program 8 downloads an ASP program from the computers mass storage device to the HP 4194A.

Use these BASIC programs to backup your library of ASP programs and to expand your uses for the HP 4194A.

```
10 1 4194A ASP DOWNLOAD PROGRAM
20 DIM Prog$(300)(100)
30 DIM S$(20),Name$(10)
40 Add=717
50 Falcon=717
60 !
70 BEEP
80 INPUT "Input File Name to be downloaded",Name$
90 ASSIGN @File TO Name$
100 I=1
110 ENTER @File,I:Prog$(I)
120 PRINT Prog$(I)
130 IF Prog$(I)="Eof" THEN 160
140 I=I+1
150 GOTO 110
160 L=I-1
170 !
180 BEEP
190 DISP "Ready to SCRATCH, and press CONT"
200 PAUSE
210 OUTPUT Add;"SCRATCH"
220 FOR I=1 TO L
230 OUTPUT Add:Prog$(I)
240 NEXT I
250 OUTPUT Add;"QUIT"
260 BEEP
270 INPUT "Input ASP file number to be stored",Asp_no$
280 BEEP
290 INPUT "Input program comment",Comment$
300 OUTPUT Add;"STORE"&Asp_no$&"',"&Comment$&"'"
310 BEEP
320 DISP "Down Load Complete"
330 END
```

Program 8 BASIC Program for ASP Download

Instruments used in this Application Note

HP 4194A Impedance/Gain-Phase Analyzer
HP 9000 Series 300 Model 310 Computer
HP 9122D 3.5 inches Floppy Disc Drive
HP 2225A Think Jet Printer

For more information, call your local HP sales office listed in the telephone directory white pages. Ask for the Electronic Instrument Department, or write to Hewlett-Packard: U.S.A. - P.O. Box 10301, Palo Alto, CA 94303-0890. Europe - Hewlett-Packard S.A., P.O. Box 529, 1180 AM Amstelveen, The Netherlands. Canada - 6877 Goreway Drive, Mississauga, L4V 1M8, Ontario. Japan - Yokogawa-Hewlett-Packard Ltd., 3-29-21, Takaido-Higashi, Suginami-ku, Tokyo 168. Far East - Hewlett-Packard Asia Headquarters, 47/F China Resources Building, 26 Harbour Road, Wanchai Hong Kong. Australasia - Hewlett-Packard Australia Ltd., 31-41 Joseph Street, Blackburn, Victoria 3130 Australia. Latin America - Hewlett-Packard Latin America Headquarters, 3495 Deer Creek Rd., Palo Alto, CA 94304. For all other areas, please write to: Hewlett-Packard Intercontinental Headquarters, 3495 Deer Creek Rd., Palo Alto, CA 94304.